Jackson 7.22

Part a.

The imaginary part of ϵ/ϵ_0 for this problem is given by

In[16]:=
```
(Imϵ[ω_] = λ (HeavisideTheta[ω - ω1] - HeavisideTheta[ω - ω2])) // TraditionalForm
```

Out[16]//TraditionalForm=

$$\lambda(\theta(\omega-\omega 1)-\theta(\omega-\omega 2))$$

Define some assumptions before we do the integral

In[17]:=
```
$Assumptions = ω ∈ Reals && λ ∈ Reals &&
   ω1 ∈ Reals && ω2 ∈ Reals && ω1 > 0 && ω2 > 0 && ω ≥ ω2 && ω1 < ω2;
```

The Kramers-Kronig relation given by Jackson 7.120 is

In[18]:=
```
(Reϵ = 1 + 2/π Integrate[ω0 Imϵ[ω0]/(ω0^2 - ω^2), {ω0, 0, Infinity}, PrincipalValue → True] //
    FullSimplify) // TraditionalForm
```

Out[18]//TraditionalForm=

$$\frac{\lambda \log\left(\frac{\omega^2-\omega 2^2}{\omega^2-\omega 1^2}\right)}{\pi}+1$$

In[19]:=
```
Plot[{Imϵ[ω], Reϵ} /. ω1 → 1 /. ω2 → 2 /. λ → 1, {ω, 0, 3}]
```

Out[19]=



Part b.

The imaginary part of ϵ/ϵ_0 for this problem is given by

In[20]:=
```
(Imϵ[ω_] = λ γ ω/((ω1^2 - ω^2)^2 + γ^2 ω^2)) // TraditionalForm;
```

Define some assumptions

In[21]:=
```
$Assumptions =
  ω ∈ Reals && λ ∈ Reals && γ ∈ Reals && ω1 ∈ Reals && ω1 > 0 && γ > 2 ω1 && ω > 0;
```

Perform partial-fraction decomposition to perform the integral

In[22]:=
```
(integrand = Apart[Factor[ω0 Imϵ[ω0]/(ω0^2 - ω^2), Extension → I]]) // TraditionalForm
```

Out[22]//TraditionalForm=

$$\frac{i \lambda \omega 0}{2 (\omega 0^2 - \omega^2) (i \gamma \omega 0 - \omega 0^2 + \omega 1^2)} - \frac{i \lambda \omega 0}{2 (\omega 0^2 - \omega^2) (-i \gamma \omega 0 - \omega 0^2 + \omega 1^2)}$$

The Kramers-Kronig relation given by Jackson 7.120 is

In[23]:=
```
(Reϵ = 1 + 2/π Integrate[integrand, {ω0, 0, Infinity}, PrincipalValue → True] /. ω1 → ω0 //
   FullSimplify) // TraditionalForm
```

Out[23]//TraditionalForm=

$$\frac{\lambda (\omega 0^2 - \omega^2)}{\gamma^2 \omega^2 + (\omega^2 - \omega 0^2)^2} + 1$$

In[29]:=
```
Plot[{Imϵ[ω], Reϵ} /. ω0 → 1 /. ω1 → 1 /. γ → 2 /. λ → 1, {ω, 0, 3}]
```

Out[29]=